Problem:
Score : 400 points

Problem Statement

We have N points in a two-dimensional plane.

The coordinates of the i-th point (1 \leq i \leq N) are (x_i,y_i).

Let us consider a rectangle whose sides are parallel to the coordinate axes that contains K or more of the N points in its interior.

Here, points on the sides of the rectangle are considered to be in the interior.

Find the minimum possible area of such a rectangle.

Constraints

2 \leq K \leq N \leq 50

-10^9 \leq x_i,y_i \leq 10^9 (1 \leq i \leq N)

x_i≠x_j (1 \leq i<j \leq N)

y_i≠y_j (1 \leq i<j \leq N)

All input values are integers. (Added at 21:50 JST)

Input

Input is given from Standard Input in the following format:

N K
x_1 y_1
:
x_{N} y_{N}

Output

Print the minimum possible area of a rectangle that satisfies the condition.

Sample Input 1

4 4
1 4
3 3
6 2
8 1

Sample Output 1

21

One rectangle that satisfies the condition with the minimum possible area has the following vertices: (1,1), (8,1), (1,4) and (8,4).

Its area is (8-1) × (4-1) = 21.

Sample Input 2

4 2
0 0
1 1
2 2
3 3

Sample Output 2

1

Sample Input 3

4 3
-1000000000 -1000000000
1000000000 1000000000
-999999999 999999999
999999999 -999999999

Sample Output 3

3999999996000000001

Watch out for integer overflows.

Solution in Haskell (GHC 7.10.3) (judge verdict: Accepted)

{-# OPTIONS_GHC -O2 -funbox-strict-fields #-}
import           Control.Monad         (replicateM)
import qualified Data.ByteString.Char8 as BC
import           Data.Int              (Int64)
import           Data.List             (minimum, sort)
import           Data.Maybe            (fromJust)

type Point = (Int, Int)
type Rect = (Point, Point)

main :: IO ()
main = do
  [n, k] <- getInts
  ps <- getPoints
  print $ solve n k ps

solve :: Int -> Int -> [Point] -> Int64
solve n k ps = minimum $ map area rs' where
  rs' = filter isValid rs
  rs  = [((x1, y1), (x2, y2)) | x1 <- xs
                              , x2 <- xs
                              , x1 < x2
                              , y1 <- ys
                              , y2 <- ys
                              , y1 < y2
                              ]
  (xs, ys) = (sort (map fst ps), sort (map snd ps))
  isValid r = length (filter (isIncluded r) ps) >= k

isIncluded :: Rect -> Point -> Bool
isIncluded ((x1, y1), (x2, y2)) (a, b) = x1 <= a && a <= x2 && y1 <= b && b <= y2

area :: Rect -> Int64
area ((x1, y1), (x2, y2)) = (x2' - x1') * (y2' - y1') where
  x1' = fromIntegral x1
  y1' = fromIntegral y1
  x2' = fromIntegral x2
  y2' = fromIntegral y2

readInts :: BC.ByteString -> [Int]
readInts = map (fst . fromJust . BC.readInt) . BC.words

getInts :: IO [Int]
getInts = readInts <$> BC.getLine

getIntss :: IO [[Int]]
getIntss = map readInts . BC.lines <$> BC.getContents

getPoints :: IO [Point]
getPoints = map (\[x, y] -> (x, y)) <$> getIntss